
1
Programmation de

contour



Programmation de contour | Fraisage d'une rainure -1226651

1.1 Fraisage d'une rainure -1226651

Programmation de contour | Fraisage d'une rainure -1226651

Données du programme

Fraisage d'une rainure Données X Y Z

Position de sécurité +150 +150 +100

Distance d'approche - - +5

Pré-position +50 +20 +100

Point de départ/final du contour +50 +20 -

Sens d'usinage Sens horaire

Données de l'outil

Ø T S F1 F2 DZ IZ

10 5 8900 1100 2000 -5 5

Ø) Diamètre

T) Numéro de l'outil

S) Vitesse de rotation

F1) Avance d'usinage

F2) Avance de retrait

DZ) Profondeur d'usinage / de perçage max.

IZ) Passe

Programmation de contour | Fraisage d'une rainure -1226651

Solution

0  BEGIN PGM 1226651 MM

1  BLK FORM 0.1 Z X+0 Y+0 Z-16

2  BLK FORM 0.2  X+100  Y+100  Z+0

3  TOOL CALL 5 Z S8900 F1100

4  L  Z+100 R0 FMAX M3

5  L  X+50  Y+20  Z+5 R0 FMAX M8

6  L  Z-5 R0 F AUTO

7  L  X+30

8  CC

9  LP  PR+40  PA+108

10 CC

11 LP  PR+40  PA+36

12 CC

13 LP  PR+40  PA-36

14 CC

15 LP  PR+40  PA-108

16 L  X+50

17 L  Z+5 R0 F2000

18 L  X+150  Y+150  Z+100 R0 FMAX

19 M30

20 END PGM 1226651 MM

Programmation de contour | Fraisage d'un contour - 1206129

1.2 Fraisage d'un contour - 1206129

Programmation de contour | Fraisage d'un contour - 1206129

Données du programme

Fraisage d'une rainure Données X Y Z

Position de sécurité +150 +150 +100

Distance d'approche - - +5

Pré-position -20 +20 +100

Point de départ/final du contour +0 +25 -

Longueur d'approche/sortie LEN20

Sens d'usinage En avalant

Données de l'outil

Ø T S F1 F2 DZ IZ

20 10 4500 1100 2000 -5 5

Ø) Diamètre

T) Numéro de l'outil

S) Vitesse de rotation

F1) Avance d'usinage

F2) Avance de retrait

DZ) Profondeur d'usinage / de perçage max.

IZ) Passe

Programmation de contour | Fraisage d'un contour - 1206129

Solution

0  BEGIN PGM 1206129 MM

1  BLK FORM 0.1 Z X+0 Y+0 Z-20

2  BLK FORM 0.2  X+100  Y+100  Z+0

3  TOOL CALL 10 Z S4500 F1100

4  L  Z+100 R0 FMAX M3

5  L  X-20  Y+20  Z+5 R0 FMAX M8

6  L  Z-5 R0 F AUTO

7  APPR LT  X+0  Y+25 LEN20 RL

8  L  X+25  Y+100

9  L  X+100  Y+75

10 L  X+75  Y+0

11 L  X+0  Y+25

12 DEP LT LEN20

13 L  Z+5 R0 F2000

14 L  X+150  Y+150  Z+100 R0 FMAX

15 M30

16 END PGM 1206129 MM

Programmation de contour | Fraisage d'un contour - 1214098

1.3 Fraisage d'un contour - 1214098

Programmation de contour | Fraisage d'un contour - 1214098

Données du programme

Fraisage d'un contour Données X Y Z

Position de sécurité +150 +150 +100

Distance d'approche - - +5

Pré-position -20 -20 +100

Point de départ/final du contour +10 +0 -

Longueur d'approche/de sortie LEN20

Sens d'usinage En avalant

Données de l'outil

Ø T S F1 F2 DZ IZ

16 8 5600 1600 2000 -5 5

Ø) Diamètre

T) Numéro de l'outil

S) Vitesse de rotation

F1) Avance d'usinage

F2) Avance de retrait

DZ) Profondeur d'usinage / de perçage max.

IZ) Passe

Programmation de contour | Fraisage d'un contour - 1214098

Solution

0  BEGIN PGM 1214098 MM

1  BLK FORM 0.1 Z X+0 Y+0 Z-16

2  BLK FORM 0.2  X+100  Y+100  Z+0

3  TOOL CALL 8 Z S5600 F1600

4  L  Z+100 R0 FMAX M3

5  L  X-20  Y-20  Z+5 R0 FMAX M8

6  L  Z-5 R0 F AUTO

7  APPR LT  X+10  Y+0 LEN20 RL

8  L  Y+14

9  CC  X+25  Y+25

10 C  X+10  Y+36 DR+

11 L  Y+100

12 DEP LT LEN20

13 L  Z+5 R0 F2000

14 L  X+150  Y+150  Z+100 R0 FMAX

15 M30

16 END PGM 1214098 MM

 1 

TNC 320 / TNC 620 / TNC 640 | Module HIT Programmation en 3 axes | 6/2018 13



Programmation de contour | Fraisage d'un contour - 1226664

1.4 Fraisage d'un contour - 1226664

Programmation de contour | Fraisage d'un contour - 1226664

Paramètres par défaut

Fraisage d'une rainure Données X Y Z

Position de sécurité +150 +150 +100

Distance d'approche - - +5

Pré-position +0 +70 +100

Point de départ/final du contour +0 +30 -

Stratégie d'approche/sortie Trajectoire circulaire avec raccor-

dement tangentiel au contour et

segment linéaire

Rayon d'approche/sortie R10

Sens d'usinage En avalant

Données de l'outil

Ø T S F1 F2 DZ IZ

20 10 4500 1700 2000 -5 5

Ø) Diamètre

T) Numéro de l'outil

S) Vitesse de rotation

F1) Avance d'usinage

F2) Avance de retrait

DZ) Profondeur d'usinage / de perçage max.

IZ) Passe

Programmation de contour | Fraisage d'un contour - 1226664

Solution

0  BEGIN PGM 1226664 MM

1  BLK FORM 0.1 Z X-50 Y-50 Z-16

2  BLK FORM 0.2  X+50  Y+50  Z+0

3  TOOL CALL 10 Z S4500 F1700

4  L  Z+100 R0 FMAX M3

5  L  X+0  Y+70  Z+5 R0 FMAX M8

6  L  Z-5 R0 F AUTO

7  APPR LCT  X+0  Y+30 R10 RL

8  FL  Y+30  AN+0

9  FC  Y+30 DR- R42.5  CCX+0  CCY+0

10 FSELECT2

11 L  X+0

12 DEP LCT  X+0  Y+70 R10

13 L  Z+5 R0 F2000

14 L  X+150  Y+150  Z+100 R0 FMAX

15 M30

16 END PGM 1226664 MM

2
Programmation de

cycles



Programmation de cycles | Fraisage de poches et de rainures - 1214155

2.1 Fraisage de poches et de rainures -

1214155

Programmation de cycles | Fraisage de poches et de rainures - 1214155

Paramètres de programme

Poches / Rainures

(Ebauche + finition)

Données X Y Z

Position de sécurité +150 +150 +100

Distance d'approche - - +5

Distance de sécurité - - +50

Surépaisseur latérale 0,2

Surépaisseur de profondeur 0,1

Facteur recouvrement 0,7

Mode fraisage En avalant

Données de l'outil

Ø T S F1 F2 DZ IZ

10 5 8900 1100 2000 -5 5

Ø) Diamètre

T) Numéro de l'outil

S) Vitesse de rotation

F1) Avance d'usinage

F2) Avance de retrait

DZ) Profondeur d'usinage / de perçage max.

IZ) Passe

Programmation de cycles | Fraisage de poches et de rainures - 1214155

Solution

0  BEGIN PGM 1214155 MM

1  BLK FORM 0.1 Z X+0 Y+0 Z-16

2  BLK FORM 0.2  X+100  Y+100  Z+0

3  TOOL CALL 5 Z S8900 F1100

4  L  Z+100 R0 FMAX M3

5  CYCL DEF 251 POCHE RECTANGULAIRE ~

Q215=+0 ;OPERATIONS D'USINAGE ~

Q218=+20 ;1ER COTE ~

Q219=+70 ;2EME COTE ~

Q220=+6 ;RAYON D'ANGLE ~

Q368=+0.2 ;SUREPAIS. LATERALE ~

Q224=+0 ;POSITION ANGULAIRE ~

Q367=+0 ;POSITION POCHE ~

Q207= AUTO ;AVANCE FRAISAGE ~

Q351=+1 ;MODE FRAISAGE ~

Q201=-5 ;PROFONDEUR ~

Q202=+5 ;PROFONDEUR DE PASSE ~

Q369=+0.1 ;SUREP. DE PROFONDEUR ~

Q206= AUTO ;AVANCE PLONGEE PROF. ~

Q338=+0 ;PASSE DE FINITION ~

Q200=+5 ;DISTANCE D'APPROCHE ~

Q203=+0 ;COORD. SURFACE PIECE ~

Q204=+50 ;SAUT DE BRIDE ~

Q370=+0.7 ;FACTEUR RECOUVREMENT ~

Q366=+2 ;PLONGEE ~

Q385= AUTO ;AVANCE DE FINITION ~

Q439=+3 ;REFERENCE AVANCE

6  L  X+30  Y+50 R0 FMAX M99

7  CYCL DEF 252 POCHE CIRCULAIRE ~

Q215=+0 ;OPERATIONS D'USINAGE ~

Q223=+30 ;DIAMETRE DU CERCLE ~

Q368=+0.2 ;SUREPAIS. LATERALE ~

Q207= AUTO ;AVANCE FRAISAGE ~

Q351=+1 ;MODE FRAISAGE ~

Q201=-5 ;PROFONDEUR ~

Q202=+5 ;PROFONDEUR DE PASSE ~

Q369=+0.1 ;SUREP. DE PROFONDEUR ~

Q206= AUTO ;AVANCE PLONGEE PROF. ~

Q338=+0 ;PASSE DE FINITION ~

Q200=+5 ;DISTANCE D'APPROCHE ~

Q203=+0 ;COORD. SURFACE PIECE ~

Q204=+50 ;SAUT DE BRIDE ~

 2 

Programmation de cycles | Fraisage de poches et de rainures - 1214155

Q370=+0.7 ;FACTEUR RECOUVREMENT ~

Q366=+1 ;PLONGEE ~

Q385= AUTO ;AVANCE DE FINITION ~

Q439=+3 ;REFERENCE AVANCE

8  L  X+75  Y+60 R0 FMAX M99

9  CYCL DEF 253 RAINURAGE ~

Q215=+0 ;OPERATIONS D'USINAGE ~

Q218=+50 ;LONGUEUR RAINURE ~

Q219=+12 ;LARGEUR RAINURE ~

Q368=+0.2 ;SUREPAIS. LATERALE ~

Q374=+20 ;POSITION ANGULAIRE ~

Q367=+0 ;POSITION RAINURE ~

Q207= AUTO ;AVANCE FRAISAGE ~

Q351=+1 ;MODE FRAISAGE ~

Q201=-5 ;PROFONDEUR ~

Q202=+5 ;PROFONDEUR DE PASSE ~

Q369=+0.1 ;SUREP. DE PROFONDEUR ~

Q206= AUTO ;AVANCE PLONGEE PROF. ~

Q338=+0 ;PASSE DE FINITION ~

Q200=+5 ;DISTANCE D'APPROCHE ~

Q203=+0 ;COORD. SURFACE PIECE ~

Q204=+50 ;SAUT DE BRIDE ~

Q366=+2 ;PLONGEE ~

Q385= AUTO ;AVANCE DE FINITION ~

Q439=+3 ;REFERENCE AVANCE

10 L  X+70  Y+22 R0 FMAX M99

11 L  X+150  Y+150  Z+100 R0 FMAX

12 M30

13 END PGM 1214155 MM

3
Techniques de

programmation



Techniques de programmation | Perçage et taraudage - 1226642

3.1 Perçage et taraudage - 1226642

Techniques de programmation | Perçage et taraudage - 1226642

Données du programme

Perçage / Filetage Données X Y Z

Position de sécurité +150 +150 +100

Distance d'approche - - +5

Distance de sécurité - - +50

Données de l'outil

Ø T S F1 F2 DZ IZ

6,8 229 6000 840 2000 -17 17

M8 264 200 - - -17 17

Ø) Diamètre

T) Numéro de l'outil

S) Vitesse de rotation

F1) Avance d'usinage

F2) Avance de retrait

DZ) Profondeur d'usinage / de perçage max.

IZ) Passe

Techniques de programmation | Perçage et taraudage - 1226642

Solution

0  BEGIN PGM 1226642 MM

1  BLK FORM 0.1 Z X+0 Y+0 Z-16

2  BLK FORM 0.2  X+100  Y+100  Z+0

3  TOOL CALL 229 Z S6000 F840

4  L  Z+100 R0 FMAX M3

5  CYCL DEF 200 PERCAGE ~

Q200=+5 ;DISTANCE D'APPROCHE ~

Q201=-17 ;PROFONDEUR ~

Q206= AUTO ;AVANCE PLONGEE PROF. ~

Q202=+17 ;PROFONDEUR DE PASSE ~

Q210=+0 ;TEMPO. EN HAUT ~

Q203=+0 ;COORD. SURFACE PIECE ~

Q204=+50 ;SAUT DE BRIDE ~

Q211=+0 ;TEMPO. AU FOND ~

Q395=+1 ;REFERENCE PROFONDEUR

6  CALL LBL 1

7  CALL LBL 2

8  L  Z+100 R0 FMAX

9  TOOL CALL 264 Z S200

10 L  Z+100 R0 FMAX M3

11 CYCL DEF 207 TARAUDAGE RIGIDE ~

Q200=+5 ;DISTANCE D'APPROCHE ~

Q201=-17 ;PROFONDEUR FILETAGE ~

Q239=+1.25 ;PAS DE VIS ~

Q203=+0 ;COORD. SURFACE PIECE ~

Q204=+50 ;SAUT DE BRIDE

12 CALL LBL 1

13 L  X+150  Y+150  Z+100 R0 FMAX

14 M30

15 LBL 1

16 CYCL DEF 221 GRILLE DE TROUS ~

Q225=+30 ;PT INITIAL 1ER AXE ~

Q226=+15 ;PT INITIAL 2EME AXE ~

Q237=+10 ;DISTANCE 1ER AXE ~

Q238=+20 ;DISTANCE 2EME AXE ~

Q242=+8 ;NOMBRE DE COLONNES ~

Q243=+2 ;NOMBRE DE LIGNES ~

Q224=+70 ;POSITION ANGULAIRE ~

Q200=+5 ;DISTANCE D'APPROCHE ~

Q203=+0 ;COORD. SURFACE PIECE ~

Q204=+50 ;SAUT DE BRIDE ~

Q301=+1 ;DEPLAC. HAUT. SECU.

Techniques de programmation | Perçage et taraudage - 1226642

17 LBL 0

18 LBL 2

19 CYCL DEF 220 CERCLE DE TROUS ~

Q216=+70 ;CENTRE 1ER AXE ~

Q217=+30 ;CENTRE 2EME AXE ~

Q244=+30 ;DIA. CERCLE PRIMITIF ~

Q245=+30 ;ANGLE INITIAL ~

Q246=+360 ;ANGLE FINAL ~

Q247=+60 ;INCREMENT ANGULAIRE ~

Q241=+6 ;NOMBRE D'USINAGES ~

Q200=+5 ;DISTANCE D'APPROCHE ~

Q203=+0 ;COORD. SURFACE PIECE ~

Q204=+50 ;SAUT DE BRIDE ~

Q301=+1 ;DEPLAC. HAUT. SECU. ~

Q365=+0 ;TYPE DEPLACEMENT

20 LBL 0

21 END PGM 1226642 MM

